

Excel Kapitel 8 Lektion 3

Exkurs WENN DANN und WENN UND

- ✘ Weitere WENN-Bedingungen
- ✘ Funktion WENN verschachteln
- ✔ Verschachteln Sie die WENN-Funktion und definieren Sie damit mehrere Kriterien. Das ist Thema dieser Lektion.



Aufgabe

Übungsdatei:  8_Weitere Funktionen

1. Weitere WENN-Bedingungen

Arbeiten Sie in der Mappe **8_Weitere Funktionen** auf der Tabelle **WENN DANN & WENN UND**. Lösen Sie zum Einstieg die Bonusberechnung mit der Wenn-Funktion von C4 bis C11.

- ▶ Die Kriterien finden Sie von F4 bis H5.
- ▶ Lösen Sie anschließend die Bonusberechnung mit der WENN-Funktion von C18 bis C25 und berechnen Sie in D18 bis D25 den Betrag.
- ▶ Die Kriterien finden Sie von F18 bis H19.

2. Funktion WENN verschachteln

Jetzt verschachteln wir die WENN-Funktion. Wir lösen die Bonusberechnung von C32 bis C39.

- ▶ Die Kriterien finden Sie von F32 bis H34.
- ▶ Verschachtelungen beginnen wir an einem Ende, zB mit dem Kriterium, dass der Umsatz aus B32 größer dem Betrag von 150 000 aus G34 ist.
`=WENN(B32>G34;`
- ▶ Kommen wir zum DANN-Argument: Wenn das Kriterium zutrifft, wollen wir 15% aus H34 ausgeben. Das fügen wir in die Funktion ein, wir haben jetzt:
`=WENN(B32>G34;H34;`
- ▶ Kommen wir zum SONST-Argument. Da wir zwei weitere Bedingungen haben, verschachteln wir und beginnen mit einer neuen WENN-Bedingung. Zunächst nur die neue WENN-Funktion:
`=WENN(B32>G33;H33;H32)`
- ▶ Die komplette Funktion lautet schlussendlich:
`=WENN(B32>=G34;H34;WENN(B32>G33;H33;H32))`
- ▶ Bitte beide Klammern schließen! Kopieren Sie die Funktion für die restlichen Zellen von C34 bis C39 und rechnen Sie den Betrag aus.

Beginnen Sie an einem Ende zu verschachteln und arbeiten Sie sich systematisch zum anderen Ende der Bedingungen durch



Übung

- Erstellen Sie in einer neuen Mappe eine Rabattliste, gestaffelt je nach dem Auftragswert:

	A	B	C	D	E
1	Verschachtelte WENN-Bedingung				
2					
3	Auftragswert	Rabatt		Kriterien Auftragswert	Rabatt
4	€ 850.000,00		unter	1000	kein Rabatt
5	€ 1.000.000,00		ab	1000	10%
6	€ 1.200.000,00		über	2000	15%
7	€ 2.800.000,00		über	5000	20%
8	€ 6.200.000,00		ab	10000	30%
9	€ 12.000.000,00				

- Berechnen Sie den Rabatt in B4 und kopieren Sie die Funktion von B5 bis B9. Formatieren Sie das Ergebnis im %-Format.

Lösung:

```
=WENN(A4>=D$8;$E$8;WENN(A4>=D$7;$E$7;WENN(A4>=D$6;$E$6;WENN(A4>=D$5;$E$5;E4))))
```

WENN ODER

- in der Mappe **8_Weitere Funktionen** auf der Tabelle **WENN DANN & WENN UND** soll von D60 bis D67 eines der beiden Kriterien zutreffen: Der Umsatz soll entweder größer / gleich sein als 100 000 *oder* der Umsatz höher / gleich sein als 45 000.

Dazu verwenden wir die WENN ODER Bedingung:

```
=WENN(ODER(B60>=H$60;C60>=H$61);H$62;H$63)
```

WENN UND verschachtelt

- Natürlich geht das auch verschachtelt. Wir beginnen wieder an einem Ende: Ab 150 000 und einem Alter von 40 gibt es 15%.

```
=WENN(UND(B73>=G$75;C73>=G$76);H$75;
```

Beim Sonst-Argument gibt es eine weitere WENN UND Bedingung: Liegt der Umsatz über 100 000 und das Alter ab 40, dann gibt es 10%, sonst 5%.

```
WENN(UND(B73>G$74;C73>=G$76);H$74;H$73))
```

Und das alles zusammen:

```
=WENN(UND(B73>=G$75;C73>=G$76);H$75;WENN(UND(B73>G$74;C73>=G$76);H$74;H$73))
```

TIPP: Das geht natürlich auch mit WENN ODER